
Краткая информация о тенологическом языке программирования.

Технологический   язык  программирования  (TL)  разработан  с   применением  простых 
лексических  и синтаксических правил,  что позволяет пользователям, не обладающим 
достаточным навыком программирования, описывать последовательности алгоритмов 
обработки параметров и последовательности технологических операций по управлению 
объектом. В то же время язык достаточно мощный, что бы не очень   ограничивать  
пользователя.

Последовательности,  написанные на языке TL, будут в дальнейшем называться  программами 
или алгоблоками.   Программы,  написанные на TL,  включаются в состав  базы данных, 
компонуются  при  его компиляции и переводятся  в объектный код,  который  затем исполняется  
интерпретатором  в микроконтроллерах.

Поскольку  вся  система   МАИС  русскоязычная,  то почти все  ключевые  слова,  используемые  
в  языке  TL,  имеют  русские синонимы.

Программы, написанные на языке TL, вводятся в базу данных при ее создании и затем могут 
быть при необходимости неоднократно изменены и модифицированы пользователем, в том 
числе, и на фоне работы системы.

Ниже представлены примеры двух алгоблоков на технологическом языке TL.  

Пример 1. 

// Начало алгоблока.
// Вычисление кода АЦП по значению аналогового значения при линейной
// шкале.
// Значение для расчета берется из текущего значения специального
// параметра.
// В полях специального параметра должно быть:
// тип обработки (tobr) - тип вводимого значения: 
//  0 - значение в мА токового диапозона k1...k2,
//  1-значение в % (0...100), 
//  2-значение шкалы (низ шкалы...верх шкалы)
// high - верх шкалы (= k2, 100 или верху требуемой шкалы)
// low - низ шкалы (= k1, 0 или низу требуемой шкалы)
// замечание: low и high в программе важны для типа обработки 2
// k1 - низ токового значения, соответствующего min коду АЦП 0, мА (4)
// k2 - верх токового значения, соответствующего max коду АЦП 4095, 
// мА (20)
// par1 - параметр типа ВЫХОД АНАЛОГОВЫЙ, которому будет передан 
// результат.

function calc_kod(special);

calc_kod()
{
    if(test(%1[ssp],0)) clear(%1[ssp],0);
    X=%1[cure];

    if(%1[tobr]=1) // в % токового диапозона
    {
        if(X < 0)  X = 0;
        if(X > 100) X = 100;
        %1[cure]=X;
        X=%1[k1] + X * (%1[k2]-%1[k1])/100;
    }
    else if(%1[tobr]=2) // пропорционально шкалы
    {
        if(X < %1[low])  X = %1[low];
        if(X > %1[high]) X = %1[high];
        %1[cure]=X;
        A=(%1[k2]-%1[k1]) / (%1[high]-%1[low]);
        B=(%1[k1]*%1[high]-%1[k2]*%1[low]) / (%1[high]-%1[low]);
        X=X*A+B;
    }
    else
    { 
        // в значениях токового диапозона
        if(X < %1[k1])  X = %1[k1];
        if(X > %1[k2]) X = %1[k2];
        %1[cure]=X;
    }

    A = 4095 / (%1[k2] - %1[k1]);
    B = -4095 * %1[k1] / (%1[k2] - %1[k1]);
    %1[par1][out:cure]  = A * X + B;
}
//*** Конец алгоблока

Пример 2.

// Начало алгоблока.
// 1БК05КЛ01 управление клапаном
//
// Входные параметры:
// '1DOU50___АВТ' - включен автоматизированный режим управления.
//
// Дискретные параметры
//  '1БК05КЛ01{'   - клапан открыт;
//  '1БК05КЛ01}'   - клапан закрыт;
//  '1БК05КЛ01МФТ' - сработала муфта клапана;
//  '1БК05КЛ01АВТ' - отключен автоматический выключатель в силовой цепи
// привода. 
//
// Лабораторные параметры
//  '1БК05КЛ01ОТК' - кнопка открытия, нажата до конца отработки команды
//  '1БК05КЛ01ЗКР' - кнопка закрытия, нажата до конца отработки команды
//  '1БК05КЛ01СТП' - кнопка стоп, нажата до конца отработки команды
//
// Параметр Выход
//  '1БК05КЛ01УПР' - управление клапаном, тип выхода 5 

// (двухпозиционный).
//
// Все параметры архивируются, время опроса параметров 0.1 сек.
// Алгоблок при каждом опросе анализирует состояние кнопок (ОТК, ЗКР,
// СТП) и неисправности оборудования:
//    1. "Сработала муфта";
//    2. "Сработал автоматический выключатель";
//    3. "Сработали одновременно сигнализаторы открытия и закрытия"
//
// Сообщения об изменениях состояний технологического оборудования
// заносятся в журнал сообщений системы: ошибки оборудования, действия
// оператора и отклик аппаратуры. Кроме того, сообщения об ошибках
// оборудования выводятся на экран.
// Кнопки управления отжимаются после выполнения команды (отображают
// состояние задвижки).
//
// При переходе в ДРУ снимаются управляющие сигналы: УПР, ОТК, ЗКР, 
// СТП и делается запись в журнал.
//

function kn01(); 
function stop();     // снятие управляющих сигналов: УПР, ОТК, ЗКР, СТП 
                     // и exit
function mess(real,real); // сообщение %1 в журнал (всегда) и на экран
                          // (при %2=1)

kn01()
{
    // Номер рабочей станции для выдачи сообщений
    L = [:station];
    // Номер рабочей станции для записи в журнал сообщений
    M = [1СИСТЕМА:stations];
    // первый опрос после загрузки
    if(test([:ssp],0)) 
    {
        [1БК05КЛ01УПР]=0;
        [1БК05КЛ01ОТК] = 0;
        [1БК05КЛ01ЗКР] = 0;
        [1БК05КЛ01СТП] = 0;
        A=0;  // предыдущее состояние муфты
        B=0;  // предыдущее состояние выключателя
        C=0;  // предыдущее состояние совместно датчиков открытия и
              // закрытия
        D=0;  //предыдущее состояние кнопки ОТКРЫТЬ
        F=0;  //предыдущее состояние кнопки ЗАКРЫТЬ  
        G=0;  //предыдущее состояние Режима Управления (ДРУ).

        exit;
    }
    // ДРУ
    if([1DOU50___АВТ]=0) 
    {  
        if (G=1) { G=0; stop(); } // "Переключение в ДРУ из АРУ"

        if([1БК05КЛ01ОТК] = 1) { [1БК05КЛ01ОТК] = 0; mess(145,1); }
        if([1БК05КЛ01ЗКР] = 1) { [1БК05КЛ01ЗКР] = 0; mess(145,1); }
        if([1БК05КЛ01СТП] = 1) { [1БК05КЛ01СТП] = 0; mess(145,1); }
        exit; 
    }

    // далее АРУ
    if (G=0) { G=1; exit;}  // "Переключение в АРУ из ДРУ"
    //отработка сигнала "Оператор выполнил команду <СТОП>."
    if([1БК05КЛ01СТП] = 1) { mess(120,0);  stop(); }

    // сообщения 1,2,3 выводятся только при первом появлении ошибки
    // отработка ош.1 "Сработала муфта."
    E=0;   //флаг ошибки
    if([1БК05КЛ01МФТ]=1) 
    {
        E=1;
        if(A=0) { A=1; mess(130,1); }  //"Сработала муфта."
    }
    else 
    {
        if(A=1) { A=0; mess(131,1); } //"Муфта в норме"
    }

    //отработка ош.2 "Сработал автоматический выключатель."
    if([1БК05КЛ01АВТ]=1)
    { 
        E=1;
        if(B=0) { B=1; mess(132,1); } //"Сработал автоматический        
                                      // выключатель."
    }
    else 
    {
        if(B=1) { B=0; mess(133,1); }   // "Автоматический выключатель
                                        // включен"
    }

    //отработка ош.3 "Сработали одновременно сигнализаторы открытия и
    // закрытия."
    if([1БК05КЛ01}]=1 & [1БК05КЛ01{]=1) 
    { 
        E=1;
        if(C=0) { C=1; mess(134,1); } //"Сработали одновременно
                                      // сигнализаторы открытия и
                                      // закрытия."
    }
    else 
    {
        if(C=1) { C=0; mess(135,1); } //"Сигнализаторы открытия и
                                      // закрытия в норме"
    }

    if(E=1) stop(); // выход при ошибках

    //установка положения кнопок в соответствии с текущим действием
    if ([1БК05КЛ01УПР]=1) 
    {
        [1БК05КЛ01ОТК]=1;
        [1БК05КЛ01ЗКР]=0;  
    }
    if ([1БК05КЛ01УПР]=-1) 
    {
        [1БК05КЛ01ОТК]=0;
        [1БК05КЛ01ЗКР]=1;  
    }

    //отработка состояния "Клапан открылся"
    if([1БК05КЛ01{]=1 & [1БК05КЛ01УПР]=1) { mess(125,1); stop(); }
    //отработка состояния "Клапан закрылся"
    else if([1БК05КЛ01}]=1 & [1БК05КЛ01УПР]=-1) 
    { mess(126,1); stop(); }
  
    //Управление возможно: подаем сигнал на открытие
    else if([1БК05КЛ01ОТК] = 1 & D=0) 
    {
        [1БК05КЛ01УПР]=1; //включаем управление открытием
        mess(121,0);      //"Оператор выполнил команду <ОТКРЫТЬ>"
        D=1;
    }
    //Управление возможно: подаем сигнал на открытие
    else if([1БК05КЛ01ЗКР] = 1 & F=0) 
    {
        [1БК05КЛ01УПР]=-1;   //включаем управление закрытием
        mess(122,0);           //"Оператор выполнил команду <ЗАКРЫТЬ>"
        F=1;
    }
}

//**********************************
stop()
{
    if([1БК05КЛ01УПР]!= 0) [1БК05КЛ01УПР]=0;
    if([1БК05КЛ01ОТК] = 1) [1БК05КЛ01ОТК] = 0;
    if([1БК05КЛ01ЗКР] = 1) [1БК05КЛ01ЗКР] = 0;
    if([1БК05КЛ01СТП] = 1) [1БК05КЛ01СТП] = 0;
    D=0; F=0;
    exit;
}

//**********************************
// %1 - номер сообщения
// %2 = 0 нет вывода на экран, = 1 есть вывод на экран
mess()  
{
    message(M, '1БК05КЛ01', %1, 1);  //журнал
    if(%2=1) message(L, '1БК05КЛ01', %1, 2);  //монитор

}
//**Конец алгоблока


